

ΑΝΑΠΤΥΞΗ ΕΦΑΡΜΟΓΩΝ

ΒΑΣΙΚΕΣ ΠΡΑΞΕΙΣ ΣΤΙΣ ΣΤΑΤΙΚΕΣ ΔΟΜΕΣ ΔΕΔΟΜΕΝΩΝ

Θέμα 1

Να αναφέρετε τα πλεονεκτήματα χρήσης πίνακα σε ένα πρόγραμμα.

Απάντηση:

- Όλα τα δεδομένα (πρέπει να είναι ίδιου τύπου) αποθηκεύονται κάτω από το ίδιο όνομα μεταβλητής διότι δε χάνονται και μπορούν να επαναχρησιμοποιηθούν αφού βρίσκονται στα στοιχεία του πίνακα.
- Η ανάγνωση όλων των δεδομένων απλοποιείται, αφού χρησιμοποιείται μόνο μία εντολή εισόδου η οποία βρίσκεται μέσα σε ένα βρόχο και επαναλαμβάνεται όσες φορές απαιτείται.

Θέμα 2

Πότε καθίσταται αναγκαία η χρήση των πινάκων σε ένα πρόγραμμα;

Απάντηση:

Η χρήση των πινάκων είναι αναγκαία αν τα δεδομένα που εισάγονται σε ένα πρόγραμμα πρέπει να διατηρούνται στη μνήμη μέχρι το τέλος της εκτέλεσης του.

Θέμα 3

Η εταιρεία "Επιστήμονες Πληροφορικής" διοργάνωσε σεμινάριο με θέμα "Επιμόρφωση Εκπαιδευτών". Το σεμινάριο παρακολούθησαν 200 εκπαιδευτικοί δευτεροβάθμιας εκπαίδευσης από 6 πόλεις της Ελλάδας (Αθήνα, Θεσσαλονίκη, Πάτρα, Λάρισα, Ηράκλειο, Ξάνθη με κωδικούς πόλης 1, 2, 3, 4, 5, 6 αντίστοιχα). Η Διεύθυνση Δευτεροβάθμιας Εκπαίδευσης ζήτησε από τη διοργανώτρια εταιρεία στοιχεία για κάθε εκπαιδευτικό όπως είναι το ονοματεπώνυμο του, το φύλο του, ο κωδικός πόλης του και τα έτη εκπαιδευτικής εμπειρίας για την έκδοση στατιστικών στοιχείων. Η Διεύθυνση ανέθεσε την επεξεργασία των στοιχείων αυτών στην εταιρεία "Statistic Software" η οποία από την πλευρά της κατασκεύασε λογισμικό το οποίο, αφού συγκέντρωσε τα παραπάνω στοιχεία σε κατάλληλες δομές δεδομένων, τύπωσε τα ακόλουθα στατιστικά στοιχεία:

- Το ποσοστό (%) των γυναικών που παρακολούθησαν το σεμινάριο.
- Τα ονοματεπώνυμα των εκπαιδευτικών που είχαν τη μεγαλύτερη εκπαιδευτική εμπειρία (θεωρήστε ότι υπάρχει τουλάχιστον ένας τέτοιος εκπαιδευτικός).
- Ποια πόλη είχε τη μικρότερη συμμετοχή στην παρακολούθηση του σεμιναρίου (θεωρήστε ότι υπάρχει μία μόνο τέτοια πόλη).

Να γράψετε αλγόριθμο κωδικοποιημένο σε "**ΓΛΩΣΣΑ**" που να υλοποιεί τις απαιτήσεις του παραπάνω λογισμικού.

Παρατηρήσεις:

1. Το όνομα κάθε πόλης και το ονοματεπώνυμο κάθε εκπαιδευτικού είναι έγκυρες αλφαριθμητικές τιμές.
2. Ο κωδικός κάθε πόλης πρέπει να είναι έγκυρη ακέραια τιμή
3. Το φύλο κάθε εκπαιδευτικού πρέπει να είναι "ΑΝΔΡΑΣ" ή "ΓΥΝΑΙΚΑ".
4. Τα έτη εκπαιδευτικής εμπειρίας είναι ακέραιες τιμές που πρέπει να ανήκουν στο διάστημα [5, 34].

Ενδεικτική Απάντηση:

ΠΡΟΓΡΑΜΜΑ Θέμα3

ΜΕΤΑΒΛΗΤΕΣ

ΠΡΑΓΜΑΤΙΚΕΣ: Κωδικός[200], Ποσοστό

ΑΚΕΡΑΙΕΣ: I, Eτη[200], Πλήθος, Max, K, J, Συμμετοχή[6], Min

ΧΑΡΑΚΤΗΡΕΣ: Ονοματεπώνυμο[200], Πόλη[6], Φύλο[200], ΟΝΕΠ[200], Πόλη_Min

ΑΡΧΗ

! Καταχώριση ονόματος κάθε πόλης σε κατάλληλη στατική δομή δεδομένων

ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 6

 ΓΡΑΨΕ 'Δώστε το όνομα της', I, 'ης πόλης:'

 ΔΙΑΒΑΣΕ Πόλη[I]

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

! Καταχώριση ον/μου, φύλου κωδ. πόλης & ετών εμπειρίας σε κατάλληλες στατικές δομές δεδομένων

ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 200

 ΓΡΑΨΕ 'Δώστε το ονοματεπώνυμο του', I, 'ου εκπαιδευτικού:'

 ΔΙΑΒΑΣΕ Ονοματεπώνυμο[I]

 ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ

 ΓΡΑΨΕ 'Δώστε το φύλο του', I, 'ου εκπαιδευτικού:'

 ΔΙΑΒΑΣΕ Φύλο[I]

 ΜΕΧΡΙΣ_ΟΤΟΥ Φύλο[I] = 'ΑΝΔΡΑΣ' Ή Φύλο[I] = 'ΓΥΝΑΙΚΑ'

```

ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΡΑΨΕ 'Δώστε τον κωδικό της πόλης του', I, 'ου εκπαιδευτικού:'
  ΔΙΑΒΑΣΕ Κωδικός[I]
  ΜΕΧΡΙΣ_ΟΤΟΥ Κωδικός[I] >= 1 ΚΑΙ Κωδικός[I] <= 6 ΚΑΙ Α_Μ(Κωδικός[I]) = Κωδικός[I]
ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΡΑΨΕ 'Δώστε τα έτη εκπαιδευτικής εμπειρίας του', I, 'ου εκπαιδευτικού:'
  ΔΙΑΒΑΣΕ Ετη[I]
  ΜΕΧΡΙΣ_ΟΤΟΥ Ετη[I] >= 5 ΚΑΙ Ετη[I] <= 34
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Πράξη της προσπέλασης - Υπολογισμός πλήθους γυναικών που παρακολούθησαν το σεμινάριο
Πλήθος ← 0
ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 200
  ΑΝ Φύλο[I] = 'ΓΥΝΑΙΚΑ' ΤΟΤΕ
    Πλήθος ← Πλήθος + 1
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Υπολογισμός και εμφάνιση ποσοστού γυναικών που παρακολούθησαν το σεμινάριο
Ποσοστό ← (Πλήθος / 200) * 100
ΓΡΑΨΕ Ποσοστό
! Πράξη της προσπέλασης - Εύρεση μεγαλύτερης εκπαιδευτικής εμπειρίας
Max ← Ετη[1]
ΓΙΑ I ΑΠΟ 2 ΜΕΧΡΙ 200
  ΑΝ Ετη[I] > Max ΤΟΤΕ
    Max ← Ετη[I]
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Πράξη αντιγραφής ονοματεπωνύμων με τη μεγαλύτερη εκπαιδευτική εμπειρία σε άλλη στατική δομή
Κ ← 0
ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 200
  ΑΝ Ετη[I] = Max ΤΟΤΕ
    Κ ← Κ + 1
    ΟΝΕΠ[Κ] ← Ονοματεπώνυμο[I]
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Πράξη της προσπέλασης - Εμφάνιση ονοματεπωνύμων με τη μεγαλύτερη εκπαιδευτική εμπειρία
ΓΡΑΨΕ 'Ονόματα εκπαιδευτικών με τη μεγαλύτερη εμπειρία:'
ΓΙΑ J ΑΠΟ 1 ΜΕΧΡΙ Κ
  ΓΡΑΨΕ ΟΝΕΠ[J]
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Αρχικοποίηση της στατικής δομής με το πλήθος συμμετοχών εκπαιδευτικών σε κάθε πόλη
ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 6
  Συμμετοχή[I] ← 0
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Πράξη της προσπέλασης - Ενημέρωση των κόμβων της δομής με το πλήθος συμμετοχών
ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 200
  Συμμετοχή[Κωδικός[I]] ← Συμμετοχή[Κωδικός[I]] + 1
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Πράξη της προσπέλασης - Εύρεση πόλης με τη μικρότερη συμμετοχή
Min ← Συμμετοχή[1]
Πόλη_Min ← Πόλη[1]
ΓΙΑ I ΑΠΟ 2 ΜΕΧΡΙ 6
  ΑΝ Συμμετοχή[I] < Min ΤΟΤΕ
    Min ← Συμμετοχή[I]
    Πόλη_Min ← Πόλη[I]
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ 'Πόλη με την μικρότερη συμμετοχή:', Πόλη_Min
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

```

Γενικές Παρατηρήσεις:

- Η εντολή εξόδου ΓΡΑΨΕ πριν από κάθε εντολή εισόδου ΔΙΑΒΑΣΕ είναι (με βάση τη συγκεκριμένη εκφώνηση) προαιρετική.
- Κόμβος δομής δεδομένων είναι κάθε αποθηκευμένη τιμή της δομής.
- Προσπέλαση είναι βασική πράξη κατά την οποία εξετάζεται ή τροποποιείται το περιεχόμενο του κόμβου κάθε δομής. Για παράδειγμα, κατά την εύρεση των Min, Max εξετάστηκε το περιεχόμενο κάθε κόμβου, ενώ κατά την ενημέρωση του πίνακα Συμμετοχή τροποποιείται το περιεχόμενο κάθε κόμβου.